

Finite Fields Computations in Maxima

Fabrizio Caruso caruso@dm.unipi.it
Jacopo D'Aurizio elianto84@gmail.com
Alasdair McAndrew amca01@gmail.com
Volker van Nek volkervannek@gmail.com

April, 2008 - July, 2013

This file documents a Maxima package for computations in finite fields. It is suitable for teaching and exploration. The first version of the package was based on the paper “Finite Fields Manipulations in Macsyma” by Kevin Rowley and Robert Silverman, SIGSAM 1989, but for which the source code is long gone. Meanwhile it contains lots of new features and optimizations implemented by Fabrizio Caruso and Jacopo D'Aurizio.

A full review was done in 2012 and 2013 by Volker van Nek. Most of the functions described below became core functions and some function names have been modified. If you use a version of Maxima prior to 5.31 please refer to an appropriate version of this file or alternatively load the necessary files from current sources. These are `src/numth.lisp` (all basic Galois Fields functions) and `share/contrib/gf/gf.mac` (square and cubic roots). If speed matters compile these two files and load the binaries.

In version 5.29 and later only for root computations it is necessary to load `gf.mac`.

Tests for basic computations in Galois Fields are located in `src/rtest_numth.mac`, tests for root computations in `share/contrib/gf/gf_test.mac`. Tests can be performed by `batch(<path_to_test_file>, test)`.

Getting started

All user commands are prefixed with “`gf_`”. All you need to start is to enter the parameters for your field. All fields in this package are of the form

$$\mathbb{F}_p[x]/m(x)$$

where p is a prime number and $m(x)$ is an polynomial irreducible over \mathbb{F}_p . If the degree of $m(x)$ is n , the the finite field will contain p^n elements, each element being a polynomial of degree strictly less than n , and all coefficients being in $\{0, 1, \dots, p-1\}$. Such a field is called a *finite field* or *Galois field* of order p^n , and is denoted \mathbb{F}_{p^n} . Note that although there are many different irreducible polynomials to choose from, if $m(x)$ and $n(x)$ are different polynomials irreducible over \mathbb{F}_p and of the same degree, then the fields

$$\mathbb{F}_p[x]/m(x)$$

and

$$\mathbb{F}_p[x]/n(x)$$

are isomorphic.

In these fields, addition and subtraction are performed on the coefficients modulo p , and multiplication and division modulo $m(x)$.

Given a prime number p and a polynomial $m(x)$ you can create a field by using the command “`gf_set_data(p, m(x))`”. `gf_set_data` checks that p is prime, and it also checks whether $m(x)$ is irreducible over \mathbb{F}_p . If these conditions are met, a primitive element in this field is computed and some pre-calculations are performed. Maxima returns a Lisp structure containing the fields data which is suitable for later use by “`gf_set_again(gf_data)`” if needed again (see below). Some of these data can be viewed by “`gf_info()`” and “`gf_infolist()`”.

```
(%i1) F16 : gf_set_data(2, x^4+x+1);
(%o1)                                     Structure[GF - DATA]
(%i2) gf_info()$
characteristic = 2
reduction polynomial = x^4 + x + 1
primitive element = x
nr of elements = 16
nr of units = 15
nr of primitive elements = 8
```

In case there is no irreducible polynomial $m(x)$ available it is sufficient to set an exponent instead. E.g. “`gf_set_data(2, 4)`” returns the same as “`gf_set_data(2, x^4+x+1)`”.

In addition to `gf_set_data` there is a command “`gf_minimal_set(p, m(x))`” to allow basic arithmetics without checking irreducibility and without computing a primitive element.

Having set up the field, we can now perform arithmetic on field elements:

Addition/subtraction. These are performed with the commands “`gf_add`” and “`gf_sub`”. In the particular field entered above, since all arithmetic of coefficients is performed modulo 2, addition and subtraction are equivalent:

```
(%i3) a : x^3+x;
(%o3)                                     x^3 + x
(%i4) b : x^3+x^2+1;
(%o4)                                     x^3 + x^2 + 1
(%i5) gf_add(a, b);
(%o5)                                     x^2 + x + 1
```

Multiplication. This is performed with the command “`gf_mult`”:

```
(%i6) gf_mult(a, b);
(%o6)                                     x^3 + x + 1
```

Inversion and division. The inverse of a field element $p(x)$ is the element $q(x)$ for which their product is equal to 1 (modulo $m(x)$). This is performed by “gf_inv”. In a finite field, division is defined as multiplying by the inverse; thus

$$a(x)/b(x) = a(x)(b(x))^{-1}.$$

These operations are performed with the commands “gf_inv” and “gf_div”:

```
(%i7) gf_inv(b);
(%o7)          x^2
(%i8) gf_div(a, b);
(%o8)          x^3 + x^2 + x
(%i9) gf_mult(a, gf_inv(b));
(%o9)          x^3 + x^2 + x
```

Exponentiation. To raise a field element to an integer power, use “gf_exp”:

```
(%i10) gf_exp(a, 14);
(%o10)          x^3 + x^2
(%i11) gf_exp(a, 15);
(%o11)          1
```

Random elements. Finally, a random element can be obtained with “gf_random”:

```
(%i12) makelist(gf_random(), i,1,3);
(%o12)          [x^2 + x + 1, x^2 + x, x^3]
```

Primitive elements, powers and logarithms

The non-zero elements of a finite field form a multiplicative group; a generator of this group is a *primitive element* in the field. The command “gf_primitive()” returns the already computed primitive element:

```
(%i13) gf_primitive();
(%o13)          x
```

Given that any non-zero element in the field can be expressed as a power of this primitive element, this power is the *index* of the element; its value is obtained with “gf_index”:

```
(%i14) a : x^3+x$
(%i15) gf_index(a);
(%o15)          9
(%i16) is(a = gf_exp(x, 9));
(%o16)          true
```

Since every element of the field can be represented as a polynomial

$$a_{n-1}x^{n-1} + a_{n-2}x^{n-2} + \cdots + a_2x^2 + a_1x + a_0$$

where every coefficient a_i satisfies $0 \leq a_i \leq p - 1$, a field element can also be considered as a list:

$$[a_{n-1}, a_{n-2}, \dots, a_2, a_1, a_0].$$

This list can be considered as the “digits” of a number in base p , in which the field element is equivalent to the number

$$a_{n-1}p^{n-1} + a_{n-2}p^{n-2} + \cdots + a_2p^2 + a_1p + a_0.$$

Thus every polynomial is equivalent to a number between 0 and $p^n - 1$; this number is obtained by “`gf_p2n`”. The reverse direction is given by “`gf_n2p`”.

Since every non-zero field element $a = a(x)$ is both equivalent to a number A and a power i of a primitive element e , we can create an array of powers corresponding to particular numbers. This array, `gf_powers`, which is created by `gf_make_logs`, is defined as follows: its i -th element (starting with zero) is the numerical form of the i -th power of the primitive element. Thus, if

$$a(x) \equiv A \equiv e^i$$

where e is the primitive element, then the i -th element of `gf_powers` is A . By definition we have $e^{p^n-1} = 1$.

The numbers A run over all integers from 1 to $p^n - 1$, and the powers i run over all the integers from 0 to $p^n - 1$, there is a corresponding “logarithm” array, `gf_logs`. The logarithm table may be considered to be indexed from 0 to $p^n - 1$, and its i -th element (ignoring the 0-th) is the power corresponding to that element.

The last array returned by `gf_make_logs` is `gf_zech_logs` which enables efficient addition.

```
(%i17) map(listarray, gf_make_logs());
(%o17)      [ [1, 2, 4, 8, 3, 6, 12, 11, 5, 10, 7, 14, 15, 13, 9, 1],
             [false, 0, 1, 4, 2, 8, 5, 10, 3, 14, 9, 7, 6, 13, 11, 12],
             [false, 4, 8, 14, 1, 10, 13, 9, 2, 7, 5, 12, 11, 6, 3, false] ]
(%i18) c : gf_exp(x, 4);
(%o18)      x + 1
(%i19) gf_p2n(c);
(%o19)      3
(%i20) gf_index(c);
(%o20)      4
(%i21) gf_logs[3];
(%o21)      4
(%i22) gf_powers[4];
(%o22)      3
```

The creation of the arrays `gf_logs` and `gf_powers` only has to be done once.

Logarithms. The array `gf_logs` contains the logarithm of any non-zero element with respect to the primitive element `e` of the field. The same holds for `gf_ind`. The logarithm of any element relative to the base of another can be obtained by the command “`gf_log`”:

```
(%i23) a : x^3+x$
(%i24) b : x^3+x^2+1$
(%i25) gf_log(a, b);
(%o25)                                     3
(%i26) is(a = gf_exp(b, 3));
(%o26)                                     true
```

We conclude that, in our field, $a = b^3$.

Primitive elements. A given field will have many primitive elements, and the command “gf_primitive_p” tests whether an element is primitive:

```
(%i27) gf_primitive_p(a);
(%o27)                                     false
(%i28) gf_primitive_p(b);
(%o28)                                     true
```

Order. By definition, any element a of the field will satisfy $a^{p^n-1} = 1$. The *order* of a is the lowest power m for which $a^m = 1$. It will be a factor of $p^n - 1$, and is obtained with “gf_order”:

```
(%i29) gf_order(a);
(%o29)                                     5
(%i30) gf_order(b);
(%o30)                                     15
```

Minimal polynomials

Associated with every element $a \in GF(p^n)$ is a polynomial $p(x)$ which satisfies:

1. $p(a) = 0$,
2. the coefficient of the highest power in $p(x)$ is one,
3. for any other polynomial $q(x)$ with $q(a) = 0$, $p(x)$ is a divisor of $q(x)$.

The polynomial $p(x)$ is thus, in a very strict sense, the *smallest* polynomial which has a as a root. It is the *minimal polynomial* of a . The command “gf_minimal_poly” calculates it:

```
(%i31) a : x^3+x$
(%i32) p : gf_minimal_poly(a);
(%o32)                                     z^4 + z^3 + z^2 + z + 1
```

To check this, substitute a for z in p :

```
(%i33) subst(a, z, p);
(%o33)                                     (x^3 + x)^4 + (x^3 + x)^3 + (x^3 + x)^2 + x^3 + x + 1
(%i34) gf_eval(%);
(%o34)                                     0
```

An application: the Chor-Rivest knapsack cryptosystem

The Chor-Rivest knapsack cryptosystem is the only knapsack cryptosystem which doesn't use modular arithmetic; instead it uses the arithmetic of finite fields. Although it has been broken, it is still a very good example of finite field arithmetic.

Assuming the two protagonists are Alice and Bob, and Alice wishes to set up a public key for Bob to encrypt messages to her. Alice chooses a finite field $\mathbb{F}_{p^n} = \mathbb{F}_p[x]/m(x)$, and a random primitive element $g(x)$. She then computes $a_i = \log_{g(x)}(x + i)$ for every $i \in \mathbb{F}_p$. She selects a random integer d for which $0 \leq d \leq p^n - 2$, and computes $c_i = (a_i + d) \pmod{p^n - 1}$. Her public key is the sequence c_i , with the parameters p and n .

To encrypt a message to Alice, Bob encodes the message as binary blocks of length p which contain n ones. Given one such block $M = (M_0, M_1, \dots, M_{p-1})$, Bob creates the cipher text

$$c = \sum_{i=0}^{p-1} M_i c_i \pmod{p^n - 1}$$

which he sends to Alice.

To decrypt c , Alice first computes $r = (c - nd) \pmod{p^n - 1}$, and then computes $u(x) = g(x)^r \pmod{m(x)}$. She then computes $s(x) = u(x) + m(x)$ and factors s into linear factors $x + t_i$. The t_i values are the positions of the ones in the message block M .

Actually, the complete cryptosystem also involves a permutation, which is applied to the sequence a_i to create c_i . But for this example we are just interested in the field arithmetic.

We shall choose the example given in chapter 8 of HAC, but without the permutation. Here the field is

$$GF(7^4) = \mathbb{F}_7[x]/(x^4 + 3x^3 + 5x^2 + 6x + 2)$$

and the primitive element chosen is $g(x) = 3x^3 + 3x^2 + 6$ and the random integer d is 1702.

First, Alice must compute her public key:

```
(%i35) gf_set_data(7, x^4+3*x^3+5*x^2+6*x+2)$
(%i36) g : 3*x^3+3*x^2+6$
(%i37) gf_primitive_p(g);
(%o37)                                     true
(%i38) a : makelist(gf_log(x+i, g), i,0,6);
(%o38) [1028, 1935, 2054, 1008, 379, 1780, 223]
(%i39) d : 1702$
(%i40) c : makelist(mod(a[i] + d, gf_order()), i,1,7);
(%o40) [330, 1237, 1356, 310, 2081, 1082, 1925]
```

Now Bob can encrypt a message to Alice; suppose one such block is $[1, 0, 1, 1, 0, 0, 1]$, which is a block of length 7 which contains exactly 4 ones.

```
(%i41) M : [1,0,1,1,0,0,1];
(%o41) [1, 0, 1, 1, 0, 0, 1]
(%i42) c : mod(sum(M[i] * c[i], i,1,7), gf_order());
(%o42) 1521
```

This last value is the ciphertext. Alice now needs to decrypt it:

```
(%i43) r : mod(c - gf_exponent() * d, gf_order());
(%o43) 1913
(%i44) u : gf_exp(g, r);
(%o44) x^3 + 3x^2 + 2x + 5
(%i45) s : u + gf_reduction();
(%o45) x^4 + 4x^3 + 8x^2 + 8x + 7
(%i46) gf_factor(s);
(%o46) x (x + 2) (x + 3) (x + 6)
```

The t_i values are 0, 2, 3, 6 and these are the positions of the ones in M .

Matrices

There are commands for dealing with matrices over finite fields. E.g. “`gf_invert_by_lu`” for inverting a matrix, and “`gf_matmult`” for multiplying matrices.

```
(%i47) gf_set_again(F16)$
(%i48) m : matrix([1,x^3+1], [x^2+1,x]);
(%o48)

$$\begin{pmatrix} 1 & x^3 + 1 \\ x^2 + 1 & x \end{pmatrix}$$

(%i49) m_inv : gf_invert_by_lu(m);
(%o49)

$$\begin{pmatrix} x^2 & 1 \\ x^3 + x & x \end{pmatrix}$$

(%i50) gf_matmult(m, m_inv);
(%o50)

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

```

Normal bases

Any field $GF(p^n)$ may be considered as a vector space over \mathbb{F}_p ; one basis is the set

$$\{1, x, x^2, \dots, x^{n-1}\}$$

which is called the *polynomial basis*. A *normal element* is a field element e for which the set

$$\{e, e^p, e^{p^2}, \dots, e^{p^{n-1}}\}$$

forms a basis. There are several commands for dealing with normal elements and bases. The command “`gf_random_normal()`” finds a normal element by simply picking field elements at random and testing each one for normality. Although this is a probabilistic algorithm, in practice it works very quickly:

```
(%i51) gf_set_data(2, x^10+x^3+1)$
(%i52) p : gf_random_normal();
(%o52) x^9 + x^8 + x^7 + x^6 + x^5 + x
```

The command “`gf_normal()`” is a brute force search through all field elements; in general it is slower than `gf_random_normal()`.

Having found a normal element the command “`gf_normal_basis()`” produces a matrix the rows of which are the coefficients of the basis elements e^{p^k} . This command takes an optional parameter; a polynomial p . If present, `gf_normal_basis()` checks if the field element is normal, and if so, produces the matrix, otherwise prints an error message. If the parameter is not given, `gf_normal_basis()` first finds a normal element, and then uses that element to produce the matrix.

With the normal basis, the command “`gf_normal_basis_rep(p, mat)`” produces the normal basis representation of p , with respect to the basis `mat`, as a list of coefficients. One attraction of using normal bases is that much arithmetic can be simplified; for example, in a normal basis representation, a power of the prime p is equivalent to a shift of coefficients:

```
(%i53) m : gf_normal_basis(p)$
(%i54) a : gf_random();
(%o54)           $x^9 + x^5 + x^3 + x^2 + 1$ 
(%i55) gf_normal_basis_rep(a, m);
(%o55)          [1, 0, 1, 0, 0, 1, 1, 0, 0, 0]
(%i56) gf_normal_basis_rep(gf_exp(a, 2), m);
(%o56)          [1, 1, 0, 1, 0, 1, 0, 1, 1, 1]
```

Large fields

`gf_set_data` computes and stores powers x^{p^k} . In case `gf_set_data` was called `gf_exp` uses the technique of modular composition. Otherwise `gf_exp` performs “repeated squaring”. `gf_index` resp. `gf_log` use a Pohlig-Hellman reduction and Brent’s version of Pollard Rho.

```
(%i57) gf_set_data(2, x^20+x^3+1);
(%o57)          Structure[GF - DATA]
(%i58) gf_info()$
characteristic = 2
reduction polynomial =  $x^{20} + x^3 + 1$ 
primitive element =  $x$ 
nr of elements = 1048576
nr of units = 1048575
nr of primitive elements = 480000
(%i59) a : x^15+x^5+1;
(%o59)           $x^{15} + x^5 + 1$ 
(%i60) index : gf_index(a);
(%o60)          720548
(%i61) gf_exp(gf_primitive(), index);
(%o61)           $x^{15} + x^5 + 1$ 
(%i62) gf_exp(a, 3^12);
(%o62)           $x^{17} + x^{16} + x^{13} + x^{12} + x^{11} + x^3 + x^2 + x$ 
```

Field extensions - the AES mixed columns operation

Above we described that an element of a finite field $GF(p^n)$ can be interpreted as a polynomial $f(x) = c_{n-1}x^{n-1} + c_{n-2}x^{n-2} + \dots + c_1x + c_0$ of degree $n - 1$ with coefficients in the prime field $GF(p)$.

An element of an extension field $GF(q^k)$ where q is a power of p , $q = p^n$, can be interpreted as a polynomial of degree $k - 1$ with coefficients in the base field $GF(p^n)$ which means that the coefficients itself can be interpreted as polynomials of degree $n - 1$ over $GF(p)$ and all coefficient arithmetic is carried out in $GF(p^n)$.

The reduction polynomial of degree k used to define $GF(q^k)$ must be irreducible over $GF(p^n)$. Otherwise the defined extension is no field and not every non-zero element is invertible.

The AES mixed columns operation is for example defined by a multiplication in $GF(256^4)$ where the non-irreducible polynomial $x^4 + 1$ is used for reduction. The element $a = 3x^3 + x^2 + x + 2$ used for the mixed columns multiplication is invertible, $a^{-1} = Bx^3 + Dx^2 + 9x + E$ in base 16, so the inverse operation is guaranteed.

The following session shows the mixed columns operation applied to one column represented by the list `[30, 5D, BF, D4]`. (These four bytes are taken from the example on page 33 of the AES specification document fips-197.pdf and they are the first four bytes that are modified by the mixed columns operation there.)

User commands for field extensions are prefixed with “`ef_`” and for nearly every `gf`-function there is a corresponding `ef`-function. The polynomial `m(x)` used for reduction can be defined by “`ef_minimal_set(m(x))`” or “`ef_set_data(m(x))`”. The `ef`-functions then refer to the previously by `gf_set_data` defined field as the base field.

```
(%i1) gf_set_data(2, x^8+x^4+x^3+x+1)$
(%i2) ef_irreducible_p(x^4+1);
(%o2)                                     false
(%i3) ef_minimal_set(x^4+1);
(%o3)          reduction polynomial = x^4 + 1
(%i4) ibase : obase : 16$
(%i5) p : ef_l2p([30, 5D, 0BF, 0D4]);
(%o5)          30 x^3 + 5D x^2 + 0BF x + 0D4
(%i6) a : 3*x^3+x^2+x+2;
(%o6)          3 x^3 + x^2 + x + 2
(%i7) ef_p2l(pa : ef_mult(p, a));
(%o7)          [0E5, 81, 66, 4]
(%i8) ai : ef_inv(a);
(%o8)          0B x^3 + 0D x^2 + 9 x + 0E
(%i9) ef_mult(ai, a);
(%o9)          1
(%i10) ef_mult(ai, pa);
(%o10)          30 x^3 + 5D x^2 + 0BF x + 0D4
(%i11) ibase : obase : 10.$
```

Square and cube roots

Multiple algorithms have been implemented in order to solve the square and cube root extraction problem over \mathbb{F}_p ; all of them basically perform an exponentiation in a extension field (ie $\mathbb{F}_{p^2} = \mathbb{F}_p[x]/(x^2 + bx + a)$ or $\mathbb{F}_{p^3} = \mathbb{F}_p[x]/(x^3 - bx - a)$) through a repeated-squaring scheme, reaching so a complexity of $O(n \log(p))$ multiplications in \mathbb{F}_p ; however, due to some differences in the representation and multiplication of elements in the extension field, they do not have exactly the same running time:

1. `msqrt(a,p)` returns the two square roots of a in \mathbb{F}_p (if they exist) representing every k -th power of x in $\mathbb{F}_p[x]/(x^2 + bx + a)$ as the first column of the matrix M^k , where M is the companion matrix associated with the polynomial $x^2 + bx + a$ and $b^2 - 4a$ is a quadratic non-residue in \mathbb{F}_p^* . It requires $5 \log_2(p)$ multiplications in \mathbb{F}_p .
2. `ssqrt(a,p)` returns the two square roots of a in \mathbb{F}_p (if they exist) using Shanks algorithm. It requires $5 \log_2(p)$ multiplications in \mathbb{F}_p .
3. `gf_sqrt(a,p)` returns the two square roots of a in \mathbb{F}_p (if they exist) using the Muller algorithm (an improved, shifted version of Cipolla-Lehmer's) and should reach the best performance, requiring only $2 \log_2(p)$ multiplications in \mathbb{F}_p .
4. `mcbert(a,p)` returns the cube roots of a in \mathbb{F}_p (if they exist) representing every k -th power of x in $\mathbb{F}_p[x]/(x^3 + bx + a)$ as the vector $(M_{2,2}, M_{2,3}, M_{3,2})$ in the matrix M^k , where M is the companion matrix associated with the polynomial $x^3 + bx + a$, irreducible over \mathbb{F}_p (Stickelberger-Redei irreducibility test for cubic polynomials is used). It requires $10 \log_2(p)$ multiplications in \mathbb{F}_p .
5. `scbert(a,p)` follows the same multiplication steps of `mcbert(a,p)`, using a simpler polynomial representation for the elements of the field extension. It requires about $11 \log_2(p)$ multiplications in \mathbb{F}_p .
6. `gf_cbert(a,p)` returns the cube roots of a in \mathbb{F}_p (if they exist) using the generalized Shanks algorithm: it's pretty fast, requiring about $4 \log_2(p)$ multiplications in \mathbb{F}_p , being so the candidate choice for cube root extraction.

Other implemented routines, using about the same ideas, are:

1. `lucas(n)`, returning the n -th Lucas number through a Muller-like scheme; it requires exactly 2 squarings and 3 sums for each bit in the binary representation of n , having so a bit-complexity bounded by $2 \log_2(n)^{3+\varepsilon}$, with ε depending on the adopted integer squaring algorithm.
2. `qsplrit(p)` and `csplrit(p)`, splitting a prime p over $\mathbb{Z}[i]$ and $\mathbb{Z}[\omega]$, ie finding (a, b) such that $p = a^2 + b^2$ (this is possible only when p is in the form $4k + 1$) or $p = a^2 + ab + b^2$ (this is possible only when p is in the form $3k + 1$), by the reduction of a binary quadratic form of the proper discriminant. They have the same complexity of the computation of a single Jacobi symbol, $O(\log(p)^2)$ bit-operations.

In Maxima 5.29 and later `lucas` is a core function.

```
(%i1) lucas(141);
(%o1) 293263001536128903730947142076
```

All the other functions listed above need to be loaded by “`load(gf)`”:

```

(%i2) load(gf)$
(%i3) msqrt(64, 1789); ssqrt(64, 1789); gf_sqrt(64, 1789);
(%o3) [1781, 8]
(%o4) [8, 1781]
(%o5) [1781, 8]
(%i6) mcbirt(64, 1789); scbirt(64, 1789); gf_cbirt(64, 1789);
(%o6) [4, 608, 1177]
(%o7) [4, 608, 1177]
(%o8) [4, 1177, 608]
(%i9) gf_factor(x^3-64, 1789);
(%o9) (x + 612) (x + 1181) (x + 1785)
(%i10) map(lambda([n], n - 1789), %);
(%o10) (x - 1177) (x - 608) (x - 4)
(%i11) qsplitt(1789);
(%o11) [5, 42]
(%i12) csplitt(1789);
(%o12) [12, 35]

```
